

---

# **Cython Reference Guide**

*Release 0.15pre*

**Stefan Behnel, Robert Bradshaw, William Stein  
Gary Furnish, Dag Seljebotn, Greg Ewing  
Gabriel Gellner, editor**

June 06, 2016



<b>1</b>	<b>Compilation</b>	<b>3</b>
1.1	From the Command Line . . . . .	3
1.2	Distutils . . . . .	4
1.3	SCons . . . . .	4
1.4	Pyximport . . . . .	4
1.5	Sage . . . . .	5
<b>2</b>	<b>Language Basics</b>	<b>7</b>
2.1	Cython File Types . . . . .	7
2.2	Declaring Data Types . . . . .	9
2.3	Statements and Expressions . . . . .	14
2.4	Functions and Methods . . . . .	16
2.5	Error and Exception Handling . . . . .	17
2.6	Conditional Compilation . . . . .	19
<b>3</b>	<b>Extension Types</b>	<b>21</b>
3.1	Attributes . . . . .	21
3.2	Methods . . . . .	22
3.3	Properties . . . . .	22
3.4	Special Methods . . . . .	23
3.5	Subclassing . . . . .	26
3.6	Forward Declarations . . . . .	26
3.7	Extension Types and None . . . . .	27
3.8	Weak Referencing . . . . .	28
3.9	External and Public Types . . . . .	28
3.10	Type Names vs. Constructor Names . . . . .	29
<b>4</b>	<b>Interfacing with Other Code</b>	<b>31</b>
4.1	C . . . . .	31
4.2	C++ . . . . .	31
4.3	Fortran . . . . .	31
4.4	Numpy . . . . .	31
<b>5</b>	<b>Special Mention</b>	<b>33</b>
<b>6</b>	<b>Limitations</b>	<b>35</b>
<b>7</b>	<b>Compiler Directives</b>	<b>37</b>

---

<b>8</b>	<b>Indices and tables</b>	<b>39</b>
8.1	Special Methods Table . . . . .	39

---

**Note:**

---

**Todo**

Most of the **boldface** is to be changed to refs or other markup later.

---

Contents:



---

## Compilation

---

- Cython code, unlike Python, must be compiled.
- This happens in two stages:
- A `.pyx` file is compiled by Cython to a `.c` file.
- The `.c` file is compiled by a C compiler to a `.so` file (or a `.pyd` file on Windows)
- The following sub-sections describe several ways to build your extension modules.

---

**Note:** The `-a` option

- Using the Cython compiler with the `-a` option will produce a really nice HTML file of the Cython generated `.c` code.
  - Double clicking on the highlighted sections will expand the code to reveal what Cython has actually generated for you.
  - This is very useful for understanding, optimizing or debugging your module.
- 

### 1.1 From the Command Line

- Run the Cython compiler command with your options and list of `.pyx` files to generate:

```
$ cython -a yourmod.pyx
```

- This creates a `yourmod.c` file. (and the `-a` switch produces a generated html file)
- Compiling your `.c` files will vary depending on your operating system.
- Python documentation for writing extension modules should have some details for your system.
- Here we give an example on a Linux system:

```
$ gcc -shared -pthread -fPIC -fwrapv -O2 -Wall -fno-strict-aliasing -I/usr/include/python2.5 -o
```

- `gcc` will need to have paths to your included header files and paths to libraries you need to link with.
- A `yourmod.so` file is now in the same directory.
- Your module, `yourmod` is available for you to import as you normally would.

## 1.2 Distutils

- Ensure Distutils is installed in your system.
- The following assumes a Cython file to be compiled called *hello.pyx*.
- Create a `setup.py` script:

```
from distutils.core import setup
from distutils.extension import Extension
from Cython.Distutils import build_ext

ext_modules = [Extension("hello", ["hello.pyx"])]

setup(
    name = 'Hello world app',
    cmdclass = {'build_ext': build_ext},
    ext_modules = ext_modules
)
```

- Run the command `python setup.py build_ext --inplace` in your system's command shell.
- Your done.. import your new extension module into your python shell or script as normal.

## 1.3 SCons

to be completed...

## 1.4 Pyximport

- For generating Cython code right in your pure python module:

```
>>> import pyximport; pyximport.install()
>>> import helloworld
Hello World
```

- Use for simple Cython builds only.
- No extra C libraries.
- No special build setup needed.
- Also has experimental compilation support for normal Python modules.
  - Allows you to automatically run Cython on every `.pyx` and `.py` module that Python imports.
  - This includes the standard library and installed packages.
  - In the case that Cython fails to compile a Python module, *pyximport* will fall back to loading the source modules instead.
- The `.py` import mechanism is installed like this:

```
>>> pyximport.install(pyximport = True)
```

---

**Note:** Authors



Paul Prescod, Stefan Behnal

---

## 1.5 Sage

The Sage notebook allows transparently editing and compiling Cython code simply by typing `%cython` at the top of a cell and evaluate it. Variables and functions defined in a Cython cell imported into the running session.

---

### Todo

Provide a link to Sage docs

---



---

## Language Basics

---

### 2.1 Cython File Types

There are three file types in cython:

- Implementation files carry a `.pyx` suffix
- Definition files carry a `.pxd` suffix
- Include files which carry a `.pxi` suffix

#### 2.1.1 Implementation File

**What can it contain?**

- Basically anything Cythonic, but see below.

**What can't it contain?**

- There are some restrictions when it comes to **extension types**, if the extension type is already defined elsewhere... **more on this later**

#### 2.1.2 Definition File

**What can it contain?**

- Any kind of C type declaration.
- `extern` C function or variable declarations.
- Declarations for module implementations.
- The definition parts of **extension types**.
- All declarations of functions, etc., for an **external library**

### What can't it contain?

- Any non-extern C variable declaration.
- Implementations of C or Python functions.
- Python class definitions
- Python executable statements.
- Any declaration that is defined as **public** to make it accessible to other Cython modules.
- This is not necessary, as it is automatic.
- a **public** declaration is only needed to make it accessible to **external C code**.

### What else?

#### cimport

- Use the **cimport** statement, as you would Python's import statement, to access these files from other definition or implementation files.
- **cimport** does not need to be called in .pyx file for .pxd file that has the same name, as they are already in the same namespace.
- For cimport to find the stated definition file, the path to the file must be appended to the `-I` option of the **cython compile command**.

#### compilation order

- When a .pyx file is to be compiled, cython first checks to see if a corresponding .pxd file exists and processes it first.

## 2.1.3 Include File

### What can it contain?

- Any Cythonic code really, because the entire file is textually embedded at the location you prescribe.

### How do I use it?

- Include the .pxi file with an `include` statement like: `include "spamstuff.pxi"`
- The `include` statement can appear anywhere in your cython file and at any indentation level
- The code in the .pxi file needs to be rooted at the “zero” indentation level.
- The included code can itself contain other `include` statements.

## 2.2 Declaring Data Types

As a dynamic language, Python encourages a programming style of considering classes and objects in terms of their methods and attributes, more than where they fit into the class hierarchy.

This can make Python a very relaxed and comfortable language for rapid development, but with a price - the ‘red tape’ of managing data types is dumped onto the interpreter. At run time, the interpreter does a lot of work searching namespaces, fetching attributes and parsing argument and keyword tuples. This run-time ‘late binding’ is a major cause of Python’s relative slowness compared to ‘early binding’ languages such as C++.

However with Cython it is possible to gain significant speed-ups through the use of ‘early binding’ programming techniques.

---

**Note:** Typing is not a necessity

Providing static typing to parameters and variables is convenience to speed up your code, but it is not a necessity. Optimize where and when needed.

---

### 2.2.1 The `cdef` Statement

The `cdef` statement is used to make C level declarations for:

#### Variables

```
cdef int i, j, k
cdef float f, g[42], *h
```

#### Structs

```
cdef struct Grail:
    int age
    float volume
```

#### Unions

```
cdef union Food:
    char *spam
    float *eggs
```

#### Enums

```
cdef enum CheeseType:
    cheddar, edam,
    camembert

cdef enum CheeseState:
    hard = 1
    soft = 2
    runny = 3
```

#### Funtions

```
cdef int eggs(unsigned long l, float f):
    ...
```

#### Extension Types

```
cdef class Spam:
    ...
```

---

**Note:** Constants

Constants can be defined by using an anonymous enum:

```
cdef enum:
    tons_of_spam = 3
```

---

## 2.2.2 Grouping cdef Declarations

A series of declarations can grouped into a cdef block:

```
cdef:
    struct Spam:
        int tons

    int i
    float f
    Spam *p

    void f(Spam *s):
        print s.tons, "Tons of spam"
```

---

**Note:** ctypedef statement

The ctypedef statement is provided for naming types:

```
ctypedef unsigned long ULong

ctypedef int *IntPtr
```

---

## 2.2.3 Parameters

- Both C and Python **function** types can be declared to have parameters C data types.
- Use normal C declaration syntax:

```
def spam(int i, char *s):
    ...

    cdef int eggs(unsigned long l, float f):
        ...
```

- As these parameters are passed into a Python declared function, they are magically **converted** to the specified C type value.
- This holds true for only numeric and string types
- If no type is specified for a parameter or a return value, it is assumed to be a Python object
  - The following takes two Python objects as parameters and returns a Python object:

```
cdef spamobjs(x, y):
    ...
```

**Note:** –

This is different then C language behavior, where it is an int by default.

- Python object types have reference counting performed according to the standard Python C-API rules:
- Borrowed references are taken as parameters
- New references are returned

### Todo

link or label here the one ref count caveat for numpy.

- The name `object` can be used to explicitly declare something as a Python Object.
- For sake of code clarity, it recomened to always use `object` explicitly in your code.
- This is also useful for cases where the name being declared would otherwise be taken for a type:

```
cdef foo(object int):
    ...
```

- As a return type:

```
cdef object foo(object int):
    ...
```

### Todo

Do a see also here ..??

## Optional Arguments

- Are supported for `cdef` and `cpdef` functions
- There differences though whether you declare them in a `.pyx` file or a `.pxd` file
  - When in a `.pyx` file, the signature is the same as it is in Python itself:

```
cdef class A:
    cdef foo(self):
        print "A"
cdef class B(A)
    cdef foo(self, x=None)
        print "B", x
cdef class C(B):
    cpdef foo(self, x=True, int k=3)
        print "C", x, k
```

- When in a `.pxd` file, the signature is different like this example: `cdef foo(x=*):`

```
cdef class A:
    cdef foo(self)
cdef class B(A)
    cdef foo(self, x=*)
cdef class C(B):
    cpdef foo(self, x=*, int k=*)
```

- The number of arguments may increase when subclassing, but the arg types and order must be the same.
- There may be a slight performance penalty when the optional arg is overridden with one that does not have default values.

## 2.2.4 Keyword-only Arguments

- As in Python 3, `def` functions can have keyword-only arguments listed after a `"*"` parameter and before a `"**"` parameter if any:

```
def f(a, b, *args, c, d = 42, e, **kwds):
    ...
```

- Shown above, the `c`, `d` and `e` arguments can not be passed as positional arguments and must be passed as keyword arguments.
- Furthermore, `c` and `e` are required keyword arguments since they do not have a default value.
- If the parameter name after the `"*"` is omitted, the function will not accept any extra positional arguments:

```
def g(a, b, *, c, d):
    ...
```

- Shown above, the signature takes exactly two positional parameters and has two required keyword parameters

## 2.2.5 Automatic Type Conversion

- For basic numeric and string types, in most situations, when a Python object is used in the context of a C value and vice versa.
- The following table summarises the conversion possibilities, assuming `sizeof(int) == sizeof(long)`:

C types	From Python types	To Python types
[unsigned] char	int, long	int
[unsigned] short		
int, long		
unsigned int	int, long	long
unsigned long		
[unsigned] long long		
float, double, long double	int, long, float	float
char *	str/bytes	str/bytes <sup>1</sup>
struct		dict

### Note: Python String in a C Context

- A Python string, passed to C context expecting a `char*`, is only valid as long as the Python string exists.
- A reference to the Python string must be kept around for as long as the C string is needed.

<sup>1</sup>The conversion is to/from str for Python 2.x, and bytes for Python 3.x.



- If this can't be guaranteed, then make a copy of the C string.
- Cython may produce an error message: Obtaining `char*` from a temporary Python value and will not resume compiling in situations like this:

```
cdef char *s
s = pystring1 + pystring2
```

- The reason is that concatenating to strings in Python produces a temporary variable.
- The variable is decref'd, and the Python string deallocated as soon as the statement has finished,
- Therefore the lvalue `"s"` is left dangling.
- The solution is to assign the result of the concatenation to a Python variable, and then obtain the `char*` from that:

```
cdef char *s
p = pystring1 + pystring2
s = p
```

---

**Note:** It is up to you to be aware of this, and not to depend on Cython's error message, as it is not guaranteed to be generated for every situation.

---

## 2.2.6 Type Casting

- The syntax used in type casting are "<" and ">"

---

**Note:** The syntax is different from C convention

---

```
cdef char *p, float *q
p = <char*>q
```

- If one of the types is a python object for `<type>x`, Cython will try and do a coercion.

---

**Note:** Cython will not stop a casting where there is no conversion, but it will emit a warning.

---

- If the address is what is wanted, cast to a `void*` first.

## Type Checking

- A cast like `<MyExtensionType>x` will cast `x` to type `MyExtensionType` without type checking at all.
- To have a cast type checked, use the syntax like: `<MyExtensionType?>x`.
- In this case, Cython will throw an error if `"x"` is not a (subclass) of `MyExtensionClass`
- Automatic type checking for extension types can be obtained by whenever `isinstance()` is used as the second parameter

## 2.2.7 Python Objects

## 2.3 Statements and Expressions

- For the most part, control structures and expressions follow Python syntax.
- When applied to Python objects, the semantics are the same unless otherwise noted.
- Most Python operators can be applied to C values with the obvious semantics.
- An expression with mixed Python and C values will have **conversions** performed automatically.
- Python operations are automatically checked for errors, with the appropriate action taken.

### 2.3.1 Differences Between Cython and C

- Most notable are C constructs which have no direct equivalent in Python.
  - An integer literal is treated as a C constant
  - It will be truncated to whatever size your C compiler thinks appropriate.
  - Cast to a Python object like this:

```
<object>10000000000000000000
```

- The "L", "LL" and the "U" suffixes have the same meaning as in C
- There is no `->` operator in Cython.. instead of `p->x`, use `p.x`.
- There is no `*` operator in Cython.. instead of `*p`, use `p[0]`.
- `&` is permissible and has the same semantics as in C.
- `NULL` is the null C pointer.
- Do NOT use 0.
- `NULL` is a reserved word in Cython
- Syntax for **Type casts** are `<type>value`.

### 2.3.2 Scope Rules

- All determination of scoping (local, module, built-in) in Cython is determined statically.
- As with Python, a variable assignment which is not declared explicitly is implicitly declared to be a Python variable residing in the scope where it was assigned.

---

**Note:**

- Module-level scope behaves the same way as a Python local scope if you refer to the variable before assigning to it.
- Tricks, like the following will NOT work in Cython:

```
try:
    x = True
except NameError:
    True = 1
```

- The above example will not work because `True` will always be looked up in the module-level scope. Do the following instead:

```
import __builtin__
try:
    True = __builtin__.True
except AttributeError:
    True = 1
```

### 2.3.3 Built-in Constants

Pre-defined Python built-in constants:

- `None`
- `True`
- `False`

### 2.3.4 Operator Precedence

- Cython uses Python precedence order, not C

### 2.3.5 For-loops

- `range()` is C optimized when the index value has been declared by `cdef`:

```
cdef i
for i in range(n):
    ...
```

- The other form available in C is the for-from style
- The target expression must be a variable name.
- The name between the lower and upper bounds must be the same as the target name.

**for i from 0 <= i < n: ...**

- Or when using a step size:

```
for i from 0 <= i < n by s:
    ...
```

- To reverse the direction, reverse the conditional operation:

```
for i from 0 >= i > n:
    ...
```

- The `break` and `continue` are permissible.
- Can contain an `else` clause.

## 2.4 Functions and Methods

- There are three types of function declarations in Cython as the sub-sections show below.
- Only “Python” functions can be called outside a Cython module from *Python interpreted code*.

### 2.4.1 Callable from Python

- Are declared with the `def` statement
- Are called with Python objects
- Return Python objects
- See **Parameters** for special consideration

### 2.4.2 Callable from C

- Are declared with the `cdef` statement.
- Are called with either Python objects or C values.
- Can return either Python objects or C values.

### 2.4.3 Callable from both Python and C

- Are declared with the `cpdef` statement.
- Can be called from anywhere, because it uses a little Cython magic.
- Uses the faster C calling conventions when being called from other Cython code.

### 2.4.4 Overriding

`cpdef` functions can override `cdef` functions:

```
cdef class A:
    cdef foo(self):
        print "A"
cdef class B(A):
    cdef foo(self, x=None)
        print "B", x
cdef class C(B):
    cpdef foo(self, x=True, int k=3)
        print "C", x, k
```

### 2.4.5 Function Pointers

- Functions declared in a `struct` are automatically converted to function pointers.
- see **using exceptions with function pointers**

## 2.4.6 Python Built-ins

The following are provided:

### Todo

incomplete

Function and arguments	Return type	Python/C API Equivalent
<code>abs(obj)</code>	object	<code>PyNumber_Absolute</code>
<code>bool(obj)</code>	object	<code>Py_True</code> , <code>Py_False</code>
<code>chr(obj)</code>	object	<code>char</code>
<code>delattr(obj, name)</code>	int	<code>PyObject_DelAttr</code>
<code>dir(obj)</code> <code>getattr(obj, name)</code> (Note 1) <code>getattr3(obj, name, default)</code>	object	<code>PyObject_Dir</code>
<code>hasattr(obj, name)</code>	int	<code>PyObject_HasAttr</code>
<code>hash(obj)</code>	int	<code>PyObject_Hash</code>
<code>intern(obj)</code>	object	<code>PyObject_InternFromString</code>
<code>isinstance(obj, type)</code>	int	<code>PyObject_IsInstance</code>
<code>issubclass(obj, type)</code>	int	<code>PyObject_IsSubclass</code>
<code>iter(obj)</code>	object	<code>PyObject_GetIter</code>
<code>len(obj)</code>	<code>Py_ssize_t</code>	<code>PyObject_Length</code>
<code>pow(x, y, z)</code> (Note 2)	object	<code>PyNumber_Power</code>
<code>reload(obj)</code>	object	<code>PyImport_ReloadModule</code>
<code>repr(obj)</code>	object	<code>PyObject_Repr</code>
<code>setattr(obj, name)</code>	void	<code>PyObject_SetAttr</code>

## 2.5 Error and Exception Handling

- A plain `cdef` declared function, that does not return a Python object...
- Has no way of reporting a Python exception to it's caller.
- Will only print a warning message and the exception is ignored.
- Inorder to propagate exceptions like this to it's caller, you need to declare an exception value for it.
- There are three forms of declaring an exception for a C compiled program.
  - First:

```
cdef int spam() except -1:
    ...
```

- In the example above, if an error occurs inside `spam`, it will immediately return with the value of `-1`, causing an exception to be propagated to it's caller.
- Functions declared with an exception value, should explicitly prevent a return of that value.
- Second:

```
cdef int spam() except? -1:
    ...
```

- Used when a `-1` may possibly be returned and is not to be considered an error.
- The `"?"` tells Cython that `-1` only indicates a *possible* error.

- Now, each time `-1` is returned, Cython generates a call to `PyErr_Occurred` to verify it is an actual error.
- Third:

```
cdef int spam() except *
```

- A call to `PyErr_Occurred` happens *every* time the function gets called.

---

**Note:** Returning `void`

A need to propagate errors when returning `void` must use this version.

---

- Exception values can only be declared for functions returning an..
- integer
- enum
- float
- pointer type
- Must be a constant expression

---

**Note:**

---

**Note:** Function pointers

- Require the same exception value specification as it's user has declared.
- Use cases here are when used as parameters and when assigned to a variable:

```
int (*grail)(int, char *) except -1
```

---

**Note:** Python Objects

- Declared exception values are **not** need.
- Remember that Cython assumes that a function without a declared return value, returns a Python object.
- Exceptions on such functions are implicitly propagated by returning `NULL`

---

**Note:** C++

- For exceptions from C++ compiled programs, see **Wrapping C++ Classes**
- 

## 2.5.1 Checking return values for non-Cython functions..

- Do not try to raise exceptions by returning the specified value.. Example:

```
cdef extern FILE *fopen(char *filename, char *mode) except NULL # WRONG!
```

- The except clause does not work that way.

- It's only purpose is to propagate Python exceptions that have already been raised by either...
- A Cython function
- A C function that calls Python/C API routines.
- To propagate an exception for these circumstances you need to raise it yourself:

```
cdef FILE *p
p = fopen("spam.txt", "r")
if p == NULL:
    raise SpamError("Couldn't open the spam file")
```

## 2.6 Conditional Compilation

- The expressions in the following sub-sections must be valid compile-time expressions.
- They can evaluate to any Python value.
- The *truth* of the result is determined in the usual Python way.

### 2.6.1 Compile-Time Definitions

- Defined using the DEF statement:

```
DEF FavouriteFood = "spam"
DEF ArraySize = 42
DEF OtherArraySize = 2 * ArraySize + 17
```

- The right hand side must be a valid compile-time expression made up of either:
  - Literal values
  - Names defined by other DEF statements
  - They can be combined using any of the Python expression syntax
  - Cython provides the following pre-defined names
    - Corresponding to the values returned by `os.uname()`
    - `UNAME_SYSNAME`
    - `UNAME_NODENAME`
    - `UNAME_RELEASE`
    - `UNAME_VERSION`
    - `UNAME_MACHINE`
- A name defined by DEF can appear anywhere an identifier can appear.
- Cython replaces the name with the literal value before compilation.
- The compile-time expression, in this case, must evaluate to a Python value of `int`, `long`, `float`, or `str`:

```
cdef int a1[ArraySize]
cdef int a2[OtherArraySize]
print "I like", FavouriteFood
```

## 2.6.2 Conditional Statements

- Similiar semantics of the C pre-processor
- The following statements can be used to conditinally include or exclude sections of code to compile.
- IF
- ELIF
- ELSE

```
IF UNAME_SYSNAME == "Windows":  
    include "icky_definitions.pxi"  
ELIF UNAME_SYSNAME == "Darwin":  
    include "nice_definitions.pxi"  
ELIF UNAME_SYSNAME == "Linux":  
    include "penguin_definitions.pxi"  
ELSE:  
    include "other_definitions.pxi"
```

- ELIF and ELSE are optional.
- IF can appear anywhere that a normal statement or declaration can appear
- It can contain any statements or declarations that would be valid in that context.
- This includes other IF and DEF statements



---

## Extension Types

---

- Normal Python as well as extension type classes can be defined.
- Extension types:
  - Are considered by Python as “built-in” types.
  - Can be used to wrap arbitrary C-data structures, and provide a Python-like interface to them from Python.
  - Attributes and methods can be called from Python or Cython code
  - Are defined by the `cdef class` statement.

```
cdef class Shrubbery:

    cdef int width, height

    def __init__(self, w, h):
        self.width = w
        self.height = h

    def describe(self):
        print "This shrubbery is", self.width, \
            "by", self.height, "cubits."
```

### 3.1 Attributes

- Are stored directly in the object’s C struct.
- Are fixed at compile time.
- You can’t add attributes to an extension type instance at run time like in normal Python.
- You can sub-class the extension type in Python to add attributes at run-time.
- There are two ways to access extension type attributes:
  - By Python look-up.
  - Python code’s only method of access.
  - By direct access to the C struct from Cython code.
  - Cython code can use either method of access, though.
- By default, extension type attributes are:

- Only accessible by direct access.
- Not accessible from Python code.
- To make attributes accessible to Python, they must be declared `public` or `readonly`:

```
cdef class Shrubbery:
    cdef public int width, height
    cdef readonly float depth
```

- The `width` and `height` attributes are readable and writable from Python code.
- The `depth` attribute is readable but not writable.

---

**Note:**

---

**Note:** You can only expose simple C types, such as ints, floats, and strings, for Python access. You can also expose Python-valued attributes.

---

---

**Note:** The `public` and `readonly` options apply only to Python access, not direct access. All the attributes of an extension type are always readable and writable by C-level access.

---

## 3.2 Methods

- `self` is used in extension type methods just like it normally is in Python.
- See **Functions and Methods**; all of which applies here.

## 3.3 Properties

- Cython provides a special syntax:

```
cdef class Spam:

    property cheese:

        "A doc string can go here."

        def __get__(self):
            # This is called when the property is read.
            ...

        def __set__(self, value):
            # This is called when the property is written.
            ...

        def __del__(self):
            # This is called when the property is deleted.
```

- The `__get__()`, `__set__()`, and `__del__()` methods are all optional.

- If they are omitted, An exception is raised when an access attempt is made.
- Below, is a full example that defines a property which can..
- Add to a list each time it is written to ("\_\_set\_\_").
- Return the list when it is read ("\_\_get\_\_").
- Empty the list when it is deleted ("\_\_del\_\_").

```
# cheesy.pyx
cdef class CheeseShop:

    cdef object cheeses

    def __cinit__(self):
        self.cheeses = []

    property cheese:

        def __get__(self):
            return "We don't have: %s" % self.cheeses

        def __set__(self, value):
            self.cheeses.append(value)

        def __del__(self):
            del self.cheeses[:]

# Test input
from cheesy import CheeseShop

shop = CheeseShop()
print shop.cheese

shop.cheese = "camembert"
print shop.cheese

shop.cheese = "cheddar"
print shop.cheese

del shop.cheese
print shop.cheese
```

```
# Test output
We don't have: []
We don't have: ['camembert']
We don't have: ['camembert', 'cheddar']
We don't have: []
```

## 3.4 Special Methods

### Note:

1. The semantics of Cython's special methods are similar in principle to that of Python's.
2. There are substantial differences in some behavior.

3. Some Cython special methods have no Python counter-part.
- 

- See the *Special Methods Table* for the many that are available.

### 3.4.1 Declaration

- Must be declared with `def` and cannot be declared with `cdef`.
- Performance is not affected by the `def` declaration because of special calling conventions

### 3.4.2 Docstrings

- Docstrings are not supported yet for some special method types.
- They can be included in the source, but may not appear in the corresponding `__doc__` attribute at run-time.
- This a Python library limitation because the `PyObject` data structure is limited

### 3.4.3 Initialization: `__cinit__()` and `__init__()`

- Any arguments passed to the extension type's constructor, will be passed to both initialization methods.
- `__cinit__()` is where you should perform C-level initialization of the object
  - This includes any allocation of C data structures.
  - **Caution** is warranted as to what you do in this method.
  - The object may not be fully valid Python object when it is called.
  - Calling Python objects, including the extensions own methods, may be hazardous.
  - By the time `__cinit__()` is called...
  - Memory has been allocated for the object.
  - All C-level attributes have been initialized to 0 or null.
  - Python have been initialized to `None`, but you can not rely on that for each occasion.
  - This initialization method is guaranteed to be called exactly once.
  - For Extensions types that inherit a base type:
    - The `__cinit__()` method of the base type is automatically called before this one.
    - The inherited `__cinit__()` method can not be called explicitly.
    - Passing modified argument lists to the base type must be done through `__init__()`.
    - It may be wise to give the `__cinit__()` method both `"*"` and `"**"` arguments.
    - Allows the method to accept or ignore additional arguments.
    - Eliminates the need for a Python level sub-class, that changes the `__init__()` method's signature, to have to override both the `__new__()` and `__init__()` methods.
    - If `__cinit__()` is declared to take no arguments except `self`, it will ignore any extra arguments passed to the constructor without complaining about a signature mis-match
- `__init__()` is for higher-level initialization and is safer for Python access.

- By the time this method is called, the extension type is a fully valid Python object.
- All operations are safe.
- This method may sometimes be called more than once, or possibly not at all.
- Take this into consideration to make sure the design of your other methods are robust of this fact.

### 3.4.4 Finalization: `__dealloc__()`

- This method is the counter-part to `__cinit__()`.
- Any C-data that was explicitly allocated in the `__cinit__()` method should be freed here.
- Use caution in this method:
- The Python object to which this method belongs may not be completely intact at this point.
- Avoid invoking any Python operations that may touch the object.
- Don't call any of this object's methods.
- It's best to just deallocate C-data structures here.
- All Python attributes of your extension type object are deallocated by Cython after the `__dealloc__()` method returns.

### 3.4.5 Arithmetic Methods

---

**Note:** Most of these methods behave differently than in Python

---

- There are not “reversed” versions of these methods... there is no `__radd__()` for instance.
- If the first operand cannot perform the operation, the same method of the second operand is called, with the operands in the same order.
- Do not rely on the first parameter of these methods, being `"self"` or the right type.
- The types of both operands should be tested before deciding what to do.
- Return `NotImplemented` for unhandled, mis-matched operand types.
- The previously mentioned points..
- Also apply to ‘in-place’ method `__ipow__()`.
- Do not apply to other ‘in-place’ methods like `__iadd__()`, in that these always take `self` as the first argument.

### 3.4.6 Rich Comparisons

---

**Note:** There are no separate methods for individual rich comparison operations.

---

- A single special method called `__richcmp__()` replaces all the individual rich compare, special method types.

- `__richcmp__()` takes an integer argument, indicating which operation is to be performed as shown in the table below.

<	0
==	2
>	4
<=	1
!=	3
>=	5

### 3.4.7 The `__next__()` Method

- Extension types used to expose an iterator interface should define a `__next__()` method.
- **Do not** explicitly supply a `next()` method, because Python does that for you automatically.

## 3.5 Subclassing

- An extension type may inherit from a built-in type or another extension type:

```
cdef class Parrot:
    ...

cdef class Norwegian(Parrot):
    ...
```

- A complete definition of the base type must be available to Cython
- If the base type is a built-in type, it must have been previously declared as an `extern` extension type.
- `cimport` can be used to import the base type, if the extern declared base type is in a `.pxd` definition file.
- In Cython, multiple inheritance is not permitted.. singular inheritance only
- Cython extension types can also be sub-classed in Python.
- Here multiple inheritance is permissible as is normal for Python.
- Even multiple extension types may be inherited, but C-layout of all the base classes must be compatible.

## 3.6 Forward Declarations

- Extension types can be “forward-declared”.
- This is necessary when two extension types refer to each other:

```
cdef class Shrubbery # forward declaration

cdef class Shrubber:
    cdef Shrubbery work_in_progress

cdef class Shrubbery:
    cdef Shrubber creator
```

- An extension type that has a base-class, requires that both forward-declarations be specified:

```
cdef class A(B)

...

cdef class A(B):
    # attributes and methods
```

## 3.7 Extension Types and None

- Parameters and C-variables declared as an Extension type, may take the value of `None`.
- This is analogous to the way a C-pointer can take the value of `NULL`.

### Note:

- Exercise caution when using `None`
- Read this section carefully.

- There is no problem as long as you are performing Python operations on it.
- This is because full dynamic type checking is applied
- When accessing an extension type's C-attributes, **make sure** it is not `None`.
- Cython does not check this for reasons of efficiency.
- Be very aware of exposing Python functions that take extension types as arguments:

```
def widen_shrubbery(Shrubbery sh, extra_width): # This is
sh.width = sh.width + extra_width
```

\* Users could **crash** the program **by** passing `None` **for** the `sh` parameter.  
 \* This could be avoided **by**:

```
def widen_shrubbery(Shrubbery sh, extra_width):
    if sh is None:
        raise TypeError
    sh.width = sh.width + extra_width
```

\* Cython provides a more convenient way **with** a `not None` clause::

```
def widen_shrubbery(Shrubbery sh not None, extra_width):
    sh.width = sh.width + extra_width
```

\* Now this function automatically checks that `sh` **is not** `None`, **as well as** that **is** the right

- `not None` can only be used in Python functions (declared with `def not cdef`).
- For `cdef` functions, you will have to provide the check yourself.
- The `self` parameter of an extension type is guaranteed to **never** be `None`.
- When comparing a value `x` with `None`, and `x` is a Python object, note the following:
  - `x is None` and `x is not None` are very efficient.
  - They translate directly to C-pointer comparisons.

- `x == None` and `x != None` or `if x:` ... (a boolean condition), will invoke Python operations and will therefore be much slower.

## 3.8 Weak Referencing

- By default, weak references are not supported.
- It can be enabled by declaring a C attribute of the object type called `__weakref__()`:

```
cdef class ExplodingAnimal:
    """This animal will self-destruct when it is
    no longer strongly referenced."""

    cdef object __weakref__
```

## 3.9 External and Public Types

### 3.9.1 Public

- When an extension type is declared `public`, Cython will generate a C-header (".h") file.
- The header file will contain the declarations for its **object-struct** and its **type-object**.
- External C-code can now access the attributes of the extension type.

### 3.9.2 External

- An `extern` extension type allows you to gain access to the internals of:
- Python objects defined in the Python core.
- Non-Cython extension modules
- The following example lets you get at the C-level members of Python's built-in "complex" object:

```
cdef extern from "complexobject.h":

    struct Py_complex:
        double real
        double imag

    ctypedef class __builtin__.complex [object PyComplexObject]:
        cdef Py_complex cval

    # A function which uses the above type
    def spam(complex c):
        print "Real:", c.cval.real
        print "Imag:", c.cval.imag
```

---

**Note:** Some important things in the example: #. `ctypedef` has been used because Python's header file has the struct declared with:



```
ctypedef struct {
    ...
} PyComplexObject;
```

1. The module of where this type object can be found is specified along side the name of the extension type. See **Implicit Importing**.
2. When declaring an external extension type...
  - Don't declare any methods, because they are Python method class the are not needed.
  - Similar to **structs** and **unions**, extension classes declared inside a `cdef extern from` block only need to declare the C members which you will actually need to access in your module.

### 3.9.3 Name Specification Clause

**Note:** Only available to **public** and **extern** extension types.

- Example:

```
[object object_struct_name, type type_object_name ]
```

- `object_struct_name` is the name to assume for the type's C-struct.
- `type_object_name` is the name to assume for the type's statically declared type-object.
- The object and type clauses can be written in any order.
- For `cdef extern from` declarations, This clause **is required**.
- The object clause is required because Cython must generate code that is compatible with the declarations in the header file.
- Otherwise the object clause is optional.
- For public extension types, both the object and type clauses **are required** for Cython to generate code that is compatible with external C-code.

### 3.10 Type Names vs. Constructor Names

- In a Cython module, the name of an extension type serves two distinct purposes:
  1. When used in an expression, it refers to a "module-level" global variable holding the type's constructor (i.e. it's type-object)
  2. It can also be used as a C-type name to declare a "type" for variables, arguments, and return values.
- Example:

```
cdef extern class MyModule.Spam:
    ...
```

- The name "Spam" serves both of these roles.
- Only "Spam" can be used as the type-name.
- The constructor can be referred to by other names.

- Upon an explicit import of “MyModule”...
- `MyModule.Spam()` could be used as the constructor call.
- `MyModule.Spam` could not be used as a type-name
- When an “as” clause is used, the name specified takes over both roles:

```
cdef extern class MyModule.Spam as Yummy:  
    ...
```

- `Yummy` becomes both type-name and a name for the constructor.
- There other ways of course, to get hold of the constructor, but `Yummy` is the only usable type-name.

---

## Interfacing with Other Code

---

### 4.1 C

### 4.2 C++

### 4.3 Fortran

### 4.4 Numpy



---

**Special Mention**

---



---

**Limitations**

---





---

## Compiler Directives

---

TODO. See <http://wiki.cython.org/enhancements/compilerdirectives>



---

## Indices and tables

---

### 8.1 Special Methods Table

This table lists all of the special methods together with their parameter and return types. In the table below, a parameter name of `self` is used to indicate that the parameter has the type that the method belongs to. Other parameters with no type specified in the table are generic Python objects.

You don't have to declare your method as taking these parameter types. If you declare different types, conversions will be performed as necessary.

#### 8.1.1 General

Name	Parameters	Return type	Description
<code>__cinit__</code>	<code>self, ...</code>		Basic initialisation (no direct Python equivalent)
<code>__init__</code>	<code>self, ...</code>		Further initialisation
<code>__dealloc__</code>	<code>self</code>		Basic deallocation (no direct Python equivalent)
<code>__cmp__</code>	<code>x, y</code>	<code>int</code>	3-way comparison
<code>__richcmp__</code>	<code>x, y, int op</code>	<code>object</code>	Rich comparison (no direct Python equivalent)
<code>__str__</code>	<code>self</code>	<code>object</code>	<code>str(self)</code>
<code>__repr__</code>	<code>self</code>	<code>object</code>	<code>repr(self)</code>
<code>__hash__</code>	<code>self</code>	<code>int</code>	Hash function
<code>__call__</code>	<code>self, ...</code>	<code>object</code>	<code>self(...)</code>
<code>__iter__</code>	<code>self</code>	<code>object</code>	Return iterator for sequence
<code>__getattr__</code>	<code>self, name</code>	<code>object</code>	Get attribute
<code>__setattr__</code>	<code>self, name, val</code>		Set attribute
<code>__delattr__</code>	<code>self, name</code>		Delete attribute

## 8.1.2 Arithmetic operators

Name	Parameters	Return type	Description
<code>__add__</code>	x, y	object	binary + operator
<code>__sub__</code>	x, y	object	binary - operator
<code>__mul__</code>	x, y	object	* operator
<code>__div__</code>	x, y	object	/ operator for old-style division
<code>__floordiv__</code>	x, y	object	// operator
<code>__truediv__</code>	x, y	object	/ operator for new-style division
<code>__mod__</code>	x, y	object	% operator
<code>__divmod__</code>	x, y	object	combined div and mod
<code>__pow__</code>	x, y, z	object	** operator or pow(x, y, z)
<code>__neg__</code>	self	object	unary - operator
<code>__pos__</code>	self	object	unary + operator
<code>__abs__</code>	self	object	absolute value
<code>__nonzero__</code>	self	int	convert to boolean
<code>__invert__</code>	self	object	~ operator
<code>__lshift__</code>	x, y	object	<< operator
<code>__rshift__</code>	x, y	object	>> operator
<code>__and__</code>	x, y	object	& operator
<code>__or__</code>	x, y	object	operator
<code>__xor__</code>	x, y	object	^ operator

## 8.1.3 Numeric conversions

Name	Parameters	Return type	Description
<code>__int__</code>	self	object	Convert to integer
<code>__long__</code>	self	object	Convert to long integer
<code>__float__</code>	self	object	Convert to float
<code>__oct__</code>	self	object	Convert to octal
<code>__hex__</code>	self	object	Convert to hexadecimal
<code>__index__</code> (2.5+ only)	self	object	Convert to sequence index

## 8.1.4 In-place arithmetic operators

Name	Parameters	Return type	Description
<code>__iadd__</code>	self, x	object	+= operator
<code>__isub__</code>	self, x	object	-= operator
<code>__imul__</code>	self, x	object	*= operator
<code>__idiv__</code>	self, x	object	/= operator for old-style division
<code>__ifloordiv__</code>	self, x	object	//= operator
<code>__itruediv__</code>	self, x	object	/= operator for new-style division
<code>__imod__</code>	self, x	object	%= operator
<code>__ipow__</code>	x, y, z	object	**= operator
<code>__ilshift__</code>	self, x	object	<<= operator
<code>__irshift__</code>	self, x	object	>>= operator
<code>__iand__</code>	self, x	object	&= operator
<code>__ior__</code>	self, x	object	= operator
<code>__ixor__</code>	self, x	object	^= operator

## 8.1.5 Sequences and mappings

Name	Parameters	Return type	Description
<code>__len__</code>	<code>self</code> <code>int</code>		<code>len(self)</code>
<code>__getitem__</code>	<code>self</code> , <code>x</code>	<code>object</code>	<code>self[x]</code>
<code>__setitem__</code>	<code>self</code> , <code>x</code> , <code>y</code>		<code>self[x] = y</code>
<code>__delitem__</code>	<code>self</code> , <code>x</code>		<code>del self[x]</code>
<code>__getslice__</code>	<code>self</code> , <code>Py_ssize_t i</code> , <code>Py_ssize_t j</code>	<code>object</code>	<code>self[i:j]</code>
<code>__setslice__</code>	<code>self</code> , <code>Py_ssize_t i</code> , <code>Py_ssize_t j</code> , <code>x</code>		<code>self[i:j] = x</code>
<code>__delslice__</code>	<code>self</code> , <code>Py_ssize_t i</code> , <code>Py_ssize_t j</code>		<code>del self[i:j]</code>
<code>__contains__</code>	<code>self</code> , <code>x</code>	<code>int</code>	<code>x in self</code>

## 8.1.6 Iterators

Name	Parameters	Return type	Description
<code>__next__</code>	<code>self</code>	<code>object</code>	Get next item (called <code>next</code> in Python)

## 8.1.7 Buffer interface

**Note:** The buffer interface is intended for use by C code and is not directly accessible from Python. It is described in the Python/C API Reference Manual under sections 6.6 and 10.6.

Name	Parameters	Return type	Description
<code>__getreadbuffer__</code>	<code>self</code> , <code>int i</code> , <code>void **p</code>		
<code>__getwritebuffer__</code>	<code>self</code> , <code>int i</code> , <code>void **p</code>		
<code>__getsegcount__</code>	<code>self</code> , <code>int *p</code>		
<code>__getcharbuffer__</code>	<code>self</code> , <code>int i</code> , <code>char **p</code>		

## 8.1.8 Descriptor objects

**Note:** Descriptor objects are part of the support mechanism for new-style Python classes. See the discussion of descriptors in the Python documentation. See also PEP 252, “Making Types Look More Like Classes”, and PEP 253, “Subtyping Built-In Types”.

Name	Parameters	Return type	Description
<code>__get__</code>	<code>self</code> , <code>instance</code> , <code>class</code>	<code>object</code>	Get value of attribute
<code>__set__</code>	<code>self</code> , <code>instance</code> , <code>value</code>		Set value of attribute
<code>__delete__</code>	<code>self</code> , <code>instance</code>		Delete attribute

- `genindex`
- `modindex`
- `search`